
2 Correction to testcase 9.2.3.3.1
2.1 f_TC_9_2_3_3_1_UTRAN()
	Object name 
	f_TC_9_2_3_3_1_UTRAN()

	Reason for change
	The current prose does not provide any guidance whether this testcase can be run in EPS only mode, or combined mode. So, the TTCN should consider both the possibilities and handle them accordingly.

In the current TTCN implementation, if the testacse is run in EPS only mode, the UE will transmit a RAU REQ at Step 23 with Update type as “RA updating”. However, the RAU ACCEPT message that is sent at Step 24 sets the updateResult to “Combined RA/LA updating”. 

This is incorrect. 

The UpdateResult should be based on what is received in the RAU REQ, not hardcoded to a specific value. This needs to be changed.

	Summary of change
	Declared  a variable v_UpdateType, and assigned it the updateType based upon what is received in the RAU REQ. Then used the variable v_UpdateType in the RAU ACCEPT message.

Please see the screenshots.

	Source of change
	NAS_TrackingArea_UG_UTRAN.ttcn

	Label
	


Before change: 

	function f_TC_9_2_3_3_1_UTRAN () runs on UTRAN_IRAT_PTC
  { // @desc First Iu mode to S1 mode intersystem change after attach; go to E-UTRAN RRC idle; RAU to UTRAN
    var START_Value v_START_Value := '00000000000000000000'B;
    var U_RLC_AM_IND v_RLC_AM_DATA_IND;
    var UTRAN_SecurityInfo_Type v_SecurityInfo;
    var PLMN_Identity v_PLMN_Identity;
    var octetstring v_PLMN;
    var octetstring v_LAC;
    var octetstring v_RAC;
    var RRCDataIndType v_ReceivedRauReq_Msg;
    var RoutingParameter v_RoutingParameter;
    var CoOrd_GUTIParam_Type v_GUTI;
    var IRAT_Coordination_MSG v_RxdMsg;
    var bitstring v_PTMSIFromLTE;
    var IRAT_Coordination_MSG v_KSIasme_octet;
    //Initialise all cells
    f_UTRAN_Init(EUTRA_UTRAN);
    v_PLMN_Identity := f_UTRAN_CellInfo_GetPLMNIdentity(utran_Cell5);
    v_PLMN := f_U_Asn2Nas_PlmnId(v_PLMN_Identity);
    v_LAC := f_UTRAN_CellInfo_GetLAC(utran_Cell5);
    v_RAC := f_UTRAN_CellInfo_GetRAC(utran_Cell5);
    //Cell started switched on with default cell power level and max attenuation
    f_UTRAN_SS_CreateCellDCH(utran_Cell5);
    //Start sending System Information
    f_UTRAN_SendDefSysInfo(utran_Cell5);
    f_UTRAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);
    f_UTRAN_TestBody_Set(true);
    //Set cell power level according to initial conditions
    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationServingCell);
    //@siclog "Step 1" siclog@
    //  Switch on UE
    f_UT_SwitchOnUE (UT, false);
    //@siclog "Step 2 - 9" siclog@
    f_UTRAN_GMMOnly_Attach(utran_Cell5);
    //@siclog "Step 10 - 12" siclog@
    f_UTRAN_UE_64kPS_Rb_Est(utran_Cell5);
    v_SecurityInfo := f_UTRAN_Security_Get (); // @sic R5s120137, R5s120320 sic@
    // Release RRC Connection
    //@siclog "Step 13" siclog@
    f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Dch);
    // Configure cell 5 as a Suitable Neighbour Cell
    //@siclog "Step 14" siclog@
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);//Send trigger to EUTRA component
    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationSuitableNeighbourCell);
    //Configure cell 5 as Serving Cell
    //@siclog "Step 21" siclog@
    EUTRA.receive (cmr_IRAT_AuthInfo) -> value v_RxdMsg;
    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationServingCell);
    EUTRA.receive (cmr_IRAT_OctetData) -> value v_KSIasme_octet; // @sic R5s140660 sic@
    v_SecurityInfo.authKeys.AuthPS := v_RxdMsg.Authentication.AuthPS;
    v_GUTI := v_RxdMsg.Authentication.GutiParams;
    v_PTMSIFromLTE := '11'B & substr (v_GUTI.M_TMSI, 2, 6) & v_GUTI.MME_Code &  substr (v_GUTI.M_TMSI, 16, 16);
    v_RoutingParameter := substr (v_PTMSIFromLTE, 8, 10); // mapped from GUTI @sic R5s110726 sic@
    //@siclog "Step 22 Void" siclog@
    //Reconfigure SS for routing UL NAS message in the AM PCO and receive an Initial Direct Transfer Routing Area Update Request
    //@siclog "Step 23" siclog@
    f_UTRAN_SS_SetConfigRRC_RB3(utran_Cell5);
    f_UTRAN_RRC_ConnEst(utran_Cell5, -, -, false); // Establish RRC connection @sic R5s140144 sic@
    alt { // @sic R5s140144 sic@
      [] U_AM.check(receive(car_RLC_Data_Ind(utran_CellDedicated,
                                       tsc_RB2,
                                       cr_108_RRC_ConnSetupCmpl(tsc_RRC_TI, ?))))
        {
           f_UTRAN_RRC_ReceiveConnSetupCmpl(utran_Cell5);
           U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated,
                                          tsc_RB3_DCCH_RRC,
                                          cdr_InitDirectTransfer_TMSIOfSamePLMN(v_RoutingParameter))) -> value v_RLC_AM_DATA_IND;  // @sic R5s110726, R5s120707 sic@
        }
      [] U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated,
                                      tsc_RB3_DCCH_RRC,
                                      cdr_InitDirectTransfer_TMSIOfSamePLMN(v_RoutingParameter))) -> value v_RLC_AM_DATA_IND  // @sic R5s110726, R5s120707 sic@
        {
           f_UTRAN_RRC_ReceiveConnSetupCmpl(utran_Cell5);
        }
    }
    //Reconfigure SS for routing UL NAS message in the Dc PCO
    f_UTRAN_SS_RemoveConfigRRC_RB3(utran_Cell5);
    //Decode and check the Routing Area Request message
    v_ReceivedRauReq_Msg := f_U_DecodeRRCDataInd_Msg (v_RLC_AM_DATA_IND.data.aM_message.uL_DCCH_Message.message_.initialDirectTransfer.nas_Message);
    if (match(v_ReceivedRauReq_Msg, cr_U_RA_UpdReq_WithUtranSignatures (cr_GMM_UpdateType,
                                                                      cr_RAI (omit, cr_LAI(omit, v_PLMN, bit2oct(v_GUTI.MME_GroupId)), bit2oct(v_GUTI.MME_Code)), //equivalent of right shifting GUTI by 32 bits (can't do that as types are all different)
                                                                      cr_PTMSI_SignatureAny,  // can't test this fully here
                                                                      int2bit(oct2int(v_KSIasme_octet.OctetData),3), // @sic R5s140660 sic@
                                                                      bit2oct(v_PTMSIFromLTE),
                                                                      cdr_MobileIdTMSI_tlv(px_PTMSI_Def, '1A'O),
                                                                      cr_RoutingAreaId2 (v_PLMN&v_LAC&v_RAC)))
        and (match (substr(v_ReceivedRauReq_Msg.routingAreaUpdateRequest.oldPTMSI_Signature.signatureValue, 0, 1), substr (bit2oct(v_GUTI.M_TMSI), 1, 1)))
        and (match (substr(v_ReceivedRauReq_Msg.routingAreaUpdateRequest.oldPTMSI_Signature.signatureValue, 1, 2), bit2oct(v_GUTI.PTMSI_Sig)))) { // P-TMSI Sig derived from GUTI @sic R5-140761 sic@
      f_UTRAN_PreliminaryPass (__FILE__, __LINE__, "Step 23");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 23");
    }    
    v_START_Value := v_RLC_AM_DATA_IND.data.aM_message.uL_DCCH_Message.message_.initialDirectTransfer.v3a0NonCriticalExtensions.initialDirectTransfer_v3a0ext.start_Value;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value);
    //@siclog "Step 23AA - 23AB" siclog@
    f_UTRAN_GMM_Authentication();
    //Activate Security
    //@siclog "Step 23A - 23B" siclog@
    f_UTRAN_RRC_Security(utran_Cell5, true, ps_domain);
    // if Security is completed, then SECURITY COMPLETE message MUST have been sent using new key
    f_UTRAN_PreliminaryPass (__FILE__, __LINE__, "Step 23B"); //Tested implcitly
    //Transmit a ROUTING AREA UPDATE ACCEPT message (no new P-TMSI nor RAI)
    //@siclog "Step 24" siclog@
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                                 tsc_RB3,
                                 cs_U_RA_UpdAcc3(cs_GMM_UpdateResult ('001'B),
                                                   cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),
                                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                                   omit, // @sic R5s110726 sic@
                                                   v_ReceivedRauReq_Msg.routingAreaUpdateRequest.pDP_ContextStatus))); // @sic R5-133584 sic@
    //Transmit a  UE CAPABILITY ENQUIRY message requesting the UE E-UTRAN capabilty.
    // The activation time specified in the Security Mode command is the next available frame, so this time will have already gone
    //@siclog "Step 25 - 27" siclog@
    f_UTRAN_UE_CapabilityInformation (utran_Cell5); // @sic R5s150162 sic@
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 26");
    //Relsease RRC Connection
    //@siclog "Step 28" siclog@
    f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Dch);
    f_UTRAN_TestBody_Set(false);
    //Tell EUTRA Component that test case is finished
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);
    //Switch off UE and release cell
    f_UTRAN_Postamble(utran_Cell5, U1_IDLE);
  } // f_TC_9_2_3_3_1_UTRAN


After change: 

	function f_TC_9_2_3_3_1_UTRAN () runs on UTRAN_IRAT_PTC
  { // @desc First Iu mode to S1 mode intersystem change after attach; go to E-UTRAN RRC idle; RAU to UTRAN
    var START_Value v_START_Value := '00000000000000000000'B;
    var U_RLC_AM_IND v_RLC_AM_DATA_IND;
    var UTRAN_SecurityInfo_Type v_SecurityInfo;
    var PLMN_Identity v_PLMN_Identity;
    var octetstring v_PLMN;
    var octetstring v_LAC;
    var octetstring v_RAC;
    var RRCDataIndType v_ReceivedRauReq_Msg;
    var RoutingParameter v_RoutingParameter;
    var CoOrd_GUTIParam_Type v_GUTI;
    var IRAT_Coordination_MSG v_RxdMsg;
    var bitstring v_PTMSIFromLTE;
    var IRAT_Coordination_MSG v_KSIasme_octet;
    var B3_Type v_UpdateType := '000'B;//WA#9_2_3_3_1
    //Initialise all cells
    f_UTRAN_Init(EUTRA_UTRAN);
    v_PLMN_Identity := f_UTRAN_CellInfo_GetPLMNIdentity(utran_Cell5);
    v_PLMN := f_U_Asn2Nas_PlmnId(v_PLMN_Identity);
    v_LAC := f_UTRAN_CellInfo_GetLAC(utran_Cell5);
    v_RAC := f_UTRAN_CellInfo_GetRAC(utran_Cell5);
    //Cell started switched on with default cell power level and max attenuation
    f_UTRAN_SS_CreateCellDCH(utran_Cell5);
    //Start sending System Information
    f_UTRAN_SendDefSysInfo(utran_Cell5);
    f_UTRAN_InterRAT_InitialiseAuthParams();  // The UE starts in EUTRA, so wait for it to send the Auth Params
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);
    f_UTRAN_TestBody_Set(true);
    //Set cell power level according to initial conditions
    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationServingCell);
    //@siclog "Step 1" siclog@
    //  Switch on UE
    f_UT_SwitchOnUE (UT, false);
    //@siclog "Step 2 - 9" siclog@
    f_UTRAN_GMMOnly_Attach(utran_Cell5);
    //@siclog "Step 10 - 12" siclog@
    f_UTRAN_UE_64kPS_Rb_Est(utran_Cell5);
    v_SecurityInfo := f_UTRAN_Security_Get (); // @sic R5s120137, R5s120320 sic@
    // Release RRC Connection
    //@siclog "Step 13" siclog@
    f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Dch);
    // Configure cell 5 as a Suitable Neighbour Cell
    //@siclog "Step 14" siclog@
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);//Send trigger to EUTRA component
    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationSuitableNeighbourCell);
    //Configure cell 5 as Serving Cell
    //@siclog "Step 21" siclog@
    EUTRA.receive (cmr_IRAT_AuthInfo) -> value v_RxdMsg;
    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationServingCell);
    EUTRA.receive (cmr_IRAT_OctetData) -> value v_KSIasme_octet; // @sic R5s140660 sic@
    v_SecurityInfo.authKeys.AuthPS := v_RxdMsg.Authentication.AuthPS;
    v_GUTI := v_RxdMsg.Authentication.GutiParams;
    v_PTMSIFromLTE := '11'B & substr (v_GUTI.M_TMSI, 2, 6) & v_GUTI.MME_Code &  substr (v_GUTI.M_TMSI, 16, 16);
    v_RoutingParameter := substr (v_PTMSIFromLTE, 8, 10); // mapped from GUTI @sic R5s110726 sic@
    //@siclog "Step 22 Void" siclog@
    //Reconfigure SS for routing UL NAS message in the AM PCO and receive an Initial Direct Transfer Routing Area Update Request
    //@siclog "Step 23" siclog@
    f_UTRAN_SS_SetConfigRRC_RB3(utran_Cell5);
    f_UTRAN_RRC_ConnEst(utran_Cell5, -, -, false); // Establish RRC connection @sic R5s140144 sic@
    alt { // @sic R5s140144 sic@
      [] U_AM.check(receive(car_RLC_Data_Ind(utran_CellDedicated,
                                       tsc_RB2,
                                       cr_108_RRC_ConnSetupCmpl(tsc_RRC_TI, ?))))
        {
           f_UTRAN_RRC_ReceiveConnSetupCmpl(utran_Cell5);
           U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated,
                                          tsc_RB3_DCCH_RRC,
                                          cdr_InitDirectTransfer_TMSIOfSamePLMN(v_RoutingParameter))) -> value v_RLC_AM_DATA_IND;  // @sic R5s110726, R5s120707 sic@
        }
      [] U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated,
                                      tsc_RB3_DCCH_RRC,
                                      cdr_InitDirectTransfer_TMSIOfSamePLMN(v_RoutingParameter))) -> value v_RLC_AM_DATA_IND  // @sic R5s110726, R5s120707 sic@
        {
           f_UTRAN_RRC_ReceiveConnSetupCmpl(utran_Cell5);
        }
    }
    //Reconfigure SS for routing UL NAS message in the Dc PCO
    f_UTRAN_SS_RemoveConfigRRC_RB3(utran_Cell5);
    //Decode and check the Routing Area Request message
    v_ReceivedRauReq_Msg := f_U_DecodeRRCDataInd_Msg (v_RLC_AM_DATA_IND.data.aM_message.uL_DCCH_Message.message_.initialDirectTransfer.nas_Message);
    if (match(v_ReceivedRauReq_Msg, cr_U_RA_UpdReq_WithUtranSignatures (cr_GMM_UpdateType,
                                                                      cr_RAI (omit, cr_LAI(omit, v_PLMN, bit2oct(v_GUTI.MME_GroupId)), bit2oct(v_GUTI.MME_Code)), //equivalent of right shifting GUTI by 32 bits (can't do that as types are all different)
                                                                      cr_PTMSI_SignatureAny,  // can't test this fully here
                                                                      int2bit(oct2int(v_KSIasme_octet.OctetData),3), // @sic R5s140660 sic@
                                                                      bit2oct(v_PTMSIFromLTE),
                                                                      cdr_MobileIdTMSI_tlv(px_PTMSI_Def, '1A'O),
                                                                      cr_RoutingAreaId2 (v_PLMN&v_LAC&v_RAC)))
        and (match (substr(v_ReceivedRauReq_Msg.routingAreaUpdateRequest.oldPTMSI_Signature.signatureValue, 0, 1), substr (bit2oct(v_GUTI.M_TMSI), 1, 1)))
        and (match (substr(v_ReceivedRauReq_Msg.routingAreaUpdateRequest.oldPTMSI_Signature.signatureValue, 1, 2), bit2oct(v_GUTI.PTMSI_Sig)))) { // P-TMSI Sig derived from GUTI @sic R5-140761 sic@
      f_UTRAN_PreliminaryPass (__FILE__, __LINE__, "Step 23");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 23");
    }
    v_UpdateType := v_ReceivedRauReq_Msg.routingAreaUpdateRequest.updateType.updateValue;//WA#9_2_3_3_1
    v_START_Value := v_RLC_AM_DATA_IND.data.aM_message.uL_DCCH_Message.message_.initialDirectTransfer.v3a0NonCriticalExtensions.initialDirectTransfer_v3a0ext.start_Value;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value);
    //@siclog "Step 23AA - 23AB" siclog@
    f_UTRAN_GMM_Authentication();
    //Activate Security
    //@siclog "Step 23A - 23B" siclog@
    f_UTRAN_RRC_Security(utran_Cell5, true, ps_domain);
    // if Security is completed, then SECURITY COMPLETE message MUST have been sent using new key
    f_UTRAN_PreliminaryPass (__FILE__, __LINE__, "Step 23B"); //Tested implcitly
    //Transmit a ROUTING AREA UPDATE ACCEPT message (no new P-TMSI nor RAI)
    //@siclog "Step 24" siclog@
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                                 tsc_RB3,
                                 cs_U_RA_UpdAcc3(cs_GMM_UpdateResult (/*WA#9_2_3_3_1 '001'B*/v_UpdateType),
                                                   cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),
                                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                                   omit, // @sic R5s110726 sic@
                                                   v_ReceivedRauReq_Msg.routingAreaUpdateRequest.pDP_ContextStatus))); // @sic R5-133584 sic@
    //Transmit a  UE CAPABILITY ENQUIRY message requesting the UE E-UTRAN capabilty.
    // The activation time specified in the Security Mode command is the next available frame, so this time will have already gone
    //@siclog "Step 25 - 27" siclog@
    f_UTRAN_UE_CapabilityInformation (utran_Cell5); // @sic R5s150162 sic@
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 26");
    //Relsease RRC Connection
    //@siclog "Step 28" siclog@
    f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Dch);
    f_UTRAN_TestBody_Set(false);
    //Tell EUTRA Component that test case is finished
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);
    //Switch off UE and release cell
    f_UTRAN_Postamble(utran_Cell5, U1_IDLE);
  } // f_TC_9_2_3_3_1_UTRAN


