
2 Correction to testcase 6.3.7
2.1 f_TC_6_3_7_UTRAN()
	Object name 
	f_TC_6_3_7_UTRAN()

	Reason for change
	As per the initial conditions the UEs allowed CSG list should be empty. In the current TTCN this is being done by calling function f_UTRAN_Preamble_EmptyCSG_AllowedList. This function performs an ATTACH REJ for CSG IDs 1 to 4 so that none of these is present in the whitelist already. It is unnecessary to perform a ATTACH REJ on any CSG ID other than CSG ID 2 which is going to be used in the testcase. This needs to be optimized.
Also, before the postamble, the RRC Connection needs to be released.

Prose CR R5-155978 is raised.

	Summary of change
	Implemented new functions  f_UTRAN_EmptyACSGL and f_UTRAN_Remove_CSG_From_ACSGL which does an ATTACH REJ for CSG ID 2, and called it instead of f_UTRAN_Preamble_EmptyCSG_AllowedList.
Also created a new record of type CSG_IdentityList_Type of CSG_Identity.

Released the RRC Connection before the postamble.

Please see the screenshot for details.

	Source of change
	CSG_TestcasesUG_UTRAN.ttcn

	Label
	WA#


Before change: 

	<<Skipped section>>
//Insert a USIM containing default values expect service n86 (TS 31.102) is "available", but the Allowed CSG List shall be empty
    f_UT_USIM_Insert (UT, "USIM with service n86 available");
    // Empty UE Allowed CSG list in USIM
    f_UTRAN_Preamble_EmptyCSG_AllowedList(utran_Cell5);

    // Broadcast correct CSG id


After change: 

	<<Skipped sction>>
//Insert a USIM containing default values expect service n86 (TS 31.102) is "available", but the Allowed CSG List shall be empty
    f_UT_USIM_Insert (UT, "USIM with service n86 available");
    // Empty UE Allowed CSG list in USIM
    //f_UTRAN_Preamble_EmptyCSG_AllowedList(utran_Cell5);//WA#6_3_7
    f_UTRAN_EmptyACSGL(utran_Cell5, {v_CSG_Identity_Cell5});//WA#6_3_7
    // Broadcast correct CSG id

f_UTRAN_SysInfo_SetSIB3_CSG_Identity (utran_Cell5, v_CSG_Identity_Cell5);
    f_UTRAN_SysInfoModifySIB3(utran_Cell5, f_UTRAN_SysInfo_GetSIB3(utran_Cell5));
    f_UTRAN_SendAuthCoOrdParameters(EUTRA); //@sic R5s140556 sic@
    //Receive updated auth params
    f_UTRAN_InterRAT_InitialiseAuthParams();
    f_UTRAN_TestBody_Set (true);
    //@siclog "Step 0" siclog@
    f_Delay (6.0);
   //@sic R5s140372 R5s140556 sic@
    //@siclog "Step 1" siclog@
    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);
    f_UTRAN_SetCellPower(utran_Cell7, -66, -68);
    //@siclog "Step 2" siclog@
    f_UT_ManualPLMN_CSG(UT ,v_PLMN, v_CSG_Identity_Cell5, true/*WA#6_3_7*/);
    //@siclog "Step 3-12" siclog@
    f_UTRAN_RRC_ConnEst (utran_Cell5);
    f_UTRAN_GMM_RAU_MappedContext (utran_Cell5);
    f_UTRAN_RRC_ConnRel (utran_Cell5, cell_Fach_Dcch);
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); //@sic R5s140372 sic@
    //@siclog "Step 13" siclog@
    //@sic R5s140372 sic@
    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationCellOFF);
    f_UTRAN_SetCellPower(utran_Cell7, tsc_AttenuationCellOFF);
    //@sic R5s140372 sic@
    //@siclog "Step 22" siclog@
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);  //@sic R5s140448 sic@
    f_UTRAN_SetCellPower(utran_Cell7, -66, -68);
    f_UTRAN_InterRAT_InitialiseAuthParams();
    //@siclog "Step 29" siclog@
    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);
    //@siclog "Step 30-37" siclog@
    t_WaitTimer.start;
    U_TM.receive (car_RRC_ConnReq (utran_Cell5,
                                   tsc_RB0,
                                   cr_108_RRC_ConnReq_r8 (?))) -> value v_RLC_TR_DATA_IND;  //@sic R5s140372 sic@
    t_WaitTimer.stop;
    v_InitialUE_Id:= v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
    f_UTRAN_RRC_ConnEst_WithoutRRCConnectionReq (utran_Cell5, v_InitialUE_Id);
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                           tsc_RB3,
                                           cr_U_RA_UpdReqAny (cr_GMM_UpdateType))) -> value v_RRC_DataInd;
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 30");
    v_START_Value := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value);
    v_UpdateType := f_UG_RauReq_GetUpdateValueForRauAcc(v_RRC_DataInd.msg.routingAreaUpdateRequest);
    f_UTRAN_GMM_RAU_Without_RAUReq(utran_Cell5, v_START_Value, v_UpdateType, v_RRC_DataInd.msg.routingAreaUpdateRequest.pDP_ContextStatus, false); // @sic R5-133584 sic@
    f_UTRAN_TestBody_Set (false);
    // Test case body finished, so let the other PTCs know
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);
    //Release RRC Connection
    f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Fach_Dcch);
    //Clear USIM, perform switch/power off and release cells
    //@sic R5s140372 sic@
    f_UTRAN_SetCellPower(utran_Cell7, tsc_AttenuationCellOFF);
    f_UTRAN_SwitchPower_Off(utran_Cell5, U1_IDLE, NotSpecified);
    f_UT_SwitchOnUE (UT, false);
    f_UTRAN_RRC_ConnEst(utran_Cell5);
    f_UTRAN_AttachRejectCSG (utran_Cell5);
    f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Fach_Dcch);//WA#6_3_7
    // Postamble
    f_UTRAN_Postamble(utran_Cell5, U6_DEREGISTERED);
    f_UTRAN_ReleaseCell (utran_Cell7, f_UTRAN_CellInfo_GetConfigType (utran_Cell7));
  }
}


2.2 f_UTRAN_EmptyACSGL
	Object name 
	f_UTRAN_EmptyACSGL

	Reason for change
	Please see change 2.1

	Summary of change
	Created new function f_UTRAN_EmptyACSGL and called new function f_UTRAN_Remove_CSG_From_ACSGL

	Source of change
	UTRAN_CSG_Functions.ttcn

	Label
	WA#


Before change: 

	function f_UTRAN_EmptyACSGL(UTRAN_CellId_Type p_CSG_CellId := utran_Cell5,
                              CSG_IdentityList_Type p_CSG_ID_List)
    runs on UTRAN_PTC
  { /* @sic R5-153578 R5s150803 sic@ */
    var integer i;
    for (i:=0; i<lengthof(p_CSG_ID_List); i:=i+1){
      f_UTRAN_Remove_CSG_From_ACSGL(p_CSG_CellId, p_CSG_ID_List[i]);
    }
  }


2.3 f_UTRAN_Remove_CSG_From_ACSGL
	Object name 
	f_UTRAN_Remove_CSG_From_ACSGL

	Reason for change
	Please see change 2.2

	Summary of change
	New function.
Please see the screenshot for details.

	Source of change
	UTRAN_CSG_Functions.ttcn

	Label
	WA#


Before change: 

	function f_UTRAN_Remove_CSG_From_ACSGL(UTRAN_CellId_Type p_CSG_CellId,
                                          CSG_Identity p_CSG_ID)
     runs on UTRAN_PTC
  { /* @sic R5-153578 R5s150803 sic@ */
    //var integer i;
    //var CSG_Identity v_CSG_Identity;
    var  U_RLC_TM_IND v_RLC_TR_DATA_IND;  //@sic R5-150124 sic@
    timer t_GuardTimer := tsc_GuardTimePreamble_EmptyCSG;
    timer t_ResponseTimer := 5.0;
    t_GuardTimer.start;
    // Set CSG cell to serving cell
    f_UTRAN_SetCellPower(p_CSG_CellId, tsc_AttenuationServingCell);
    // Broadcast CSG Identity in Sysinfo    
    f_UTRAN_SysInfo_SetSIB3_CSG_Identity (p_CSG_CellId, p_CSG_ID);
    f_UTRAN_SysInfoModifySIB3 ( p_CSG_CellId, f_UTRAN_SysInfo_GetSIB3(p_CSG_CellId) );
    // Switch on UE
    f_UT_SwitchOnUE (UT, false);    
    // Wait 5s
    t_ResponseTimer.start;
    alt {
        // UE intiates Attach if the CSG id is included in it's Allowed CSG list
        [] U_TM.receive(car_RRC_ConnReq(p_CSG_CellId,
                                        tsc_RB0,
                                        cr_108_RRC_ConnReq_r8((registration, terminatingConversationalCall)))) -> value v_RLC_TR_DATA_IND
          {
            t_ResponseTimer.stop;
            f_UTRAN_RRC_ConnEst_WithoutRRCConnectionReq ( p_CSG_CellId, v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity );
            f_UTRAN_AttachRejectCSG ( p_CSG_CellId );
          }
        [] t_ResponseTimer.timeout
          {
            // Manual select the CSG identity
            f_UT_ManualPLMN_CSG ( UT, f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(p_CSG_CellId)), p_CSG_ID, true );
            // Attach is rejected with CSG not allowed cause
            f_UTRAN_RRC_ConnEst(p_CSG_CellId);
            f_UTRAN_AttachRejectCSG (p_CSG_CellId);
          }
      }
      //Release RRC connection
      f_UTRAN_RRC_ConnRel (p_CSG_CellId, cell_Fach_Dcch);
      // And then just need to switch UE off
      f_UT_SwitchOffUE (UT, false, CNF_REQUIRED);
      // Power off CSG cell
      f_UTRAN_SetCellPower ( p_CSG_CellId, tsc_AttenuationCellOFF );
  }


2.4 Module UTRAN_CSG_Functions
	Object name 
	CSG_IdentityList_Type

	Reason for change
	Please see Change 2.1

	Summary of change
	Implemented new record of type CSG_Identity.
Please see the screenshot for details.

	Source of change
	UTRAN_CSG_Functions.ttcn

	Label
	WA#


Before change: 

	module UTRAN_CSG_Functions {
  //=========================================================================
  // Imports
  //=========================================================================
  import from CommonDefs all;
  import from UTRAN_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from UTRAN_ASP_Definitions language "ASN.1:2002" all;
  import from UTRAN_Component all;
  import from UTRAN_CellInfo all;
  import from UTRAN_ConfigurationSteps all;
  import from UTRAN_IRAT_CommonFunctions all;
  import from UTRAN_TTCN_ASP_Templates all;
  import from UTRAN_NAS_PDU_Templates all;
  import from UTRAN_TypeDefs all;
  import from UTRAN_CommonDefs all;
  import from UTRAN_RRC_Templates all;
  import from UTRANGERAN_CommonFunctions all;
  import from UTRAN_ASN1_ASP_Templates all;
  import from UTRAN_CommonFunctions all;
  import from UpperTesterFunctions all;
  import from NAS_24008Templates_IRAT all;
  import from NAS_24008TypeDefs all;
  type record length(1..3) of CSG_Identity CSG_IdentityList_Type;//WA#6_3_7
  //==============================================================================


2.5 f_UT_ManualPLMN_CSG
	Object name 
	f_UT_ManualPLMN_CSG

	Reason for change
	As per the current implementation of this function, the user prompt is always “SELECT_CSG PLMN "00101" CSG 2”. This does not indicate whether the CSG cell is EUTRAN or UTRAN. 

Since there is no standardized method to trigger a manual CSG selection, the individual UE implementations have their own way to enforce this. 

Thus, it will be helpful to provide the RAT information in this prompt message so that the test automation can be more streamlined and effective.

It is hence proposed to provide the RAT information when the CSG selection is to be done for a UTRAN cell. 

	Summary of change
	1) Modified the function f_UT_ManualPLMN_CSG to include a Boolean flag bIsTheCellUTRAN which by default is set to false. When a UTRAN CSG Cell is to be manually selected, the funcation call for f_ManualPLMN_CSG will be passing TRUE for this flag. Based upon this flag, the appropriate prompt will be displayed. (In this case the prompt will be SELECT_UTRAN_CSG PLMN "00101" CSG 2 

2) Created a new template cs_MMI_CSG_PLMN_UTRAN which sends cs_MMI_Cmd(“SELECT_UTRAN_CSG”, p_ParamList); in the file UpperTesterSystemCmds.ttcn

3) Included SELECT_UTRAN_CSG in the enumerated Ut_Common_Req_Type in file UpperTesterPtctoMtcDefs.ttcn

4) In the function fl_UT_HandleCmd added handling for SELECT_UTRAN_CSG.
Please see the screenshot for details.

	Source of change
	UpperTesterFunctions.ttcn

	Label
	WA#


Before change: 

	function f_UT_ManualPLMN_CSG(UT_PTC_MTC_PORT p_Port,
                               NAS_PlmnId p_PLMN_ID,
                               bitstring p_CSG_ID)
  { /* Objective: To ask the operator to manually select the specified CSG cell TTCN waits until operation is confirmed before continuing */    

fl_UT_ApplyCommand(p_Port, cas_UT_Req(SELECT_CSG, NO_CNF_REQUIRED, cs_PlmnCSG(fl_ConvertPLMN(p_PLMN_ID), int2str(bit2int(p_CSG_ID)))));  // @sic R5s120298 sic@
  }


After change: 

	function f_UT_ManualPLMN_CSG(UT_PTC_MTC_PORT p_Port,
                               NAS_PlmnId p_PLMN_ID,
                               bitstring p_CSG_ID, 
                               boolean bIsTheCellUTRAN := false/*WA#6_3_7*/)
  { /* Objective: To ask the operator to manually select the specified CSG cell TTCN waits until operation is confirmed before continuing */
    if(bIsTheCellUTRAN)
    {
    
fl_UT_ApplyCommand(p_Port, cas_UT_Req(SELECT_UTRAN_CSG, NO_CNF_REQUIRED, cs_PlmnCSG(fl_ConvertPLMN(p_PLMN_ID), int2str(bit2int(p_CSG_ID)))));  // @sic R5s120298 sic@
    }
    else
    {
        fl_UT_ApplyCommand(p_Port, cas_UT_Req(SELECT_CSG, NO_CNF_REQUIRED, cs_PlmnCSG(fl_ConvertPLMN(p_PLMN_ID), int2str(bit2int(p_CSG_ID)))));  // @sic R5s120298 sic@
    }
  }


2.6 cs_MMI_CSG_PLMN_UTRAN
	Object name 
	cs_MMI_CSG_PLMN_UTRAN

	Reason for change
	Please see change 2.5 point 2 above.

	Summary of change
	New change

	Source of change
	UpperTesterSystemCmds.ttcn

	Label
	WA#


Before change: 

	//WA#6_3_7
  template (value) Ut_System_Cmd_Type cs_MMI_CSG_PLMN_UTRAN (template (value) Ut_ParameterList_Type p_ParamList) :=         /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS, SSNITZ, UTRAN) */
    cs_MMI_Cmd("SELECT_UTRAN_CSG", p_ParamList);


2.7 type enumerated Ut_Common_Req_Type
	Object name 
	Ut_Common_Req_Type

	Reason for change
	Please see change 2.5 point 3 above.

	Summary of change
	New change

	Source of change
	UpperTesterPtcToMtcDefs.ttcn

	Label
	WA#


Before change: 

	type enumerated Ut_Common_Req_Type {          /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS, SSNITZ, UTRAN) */
    ACCEPT_MTSI_TEXT,
    ACTIVATE_BEARER,
.

.

.

SELECT_UTRAN_CSG,//WA#6_3_7


2.8 fl_UT_HandleCmd
	Object name 
	fl_UT_HandleCmd

	Reason for change
	Please see change 2.5 point 4 above.

	Summary of change
	New change

	Source of change
	MTC_UpperTester.ttcn

	Label
	WA#


Before change: 

	<<SKIPPED SECTION>>
case (PLMN_AUTOMATIC)  {
        v_SystemCmd := cs_AT_PLMN_AUTOMATIC; // @sic R5-106675 sic@
      }
      case (PLMN_MANUAL) {
        v_SystemCmd := cs_AT_SET_PLMN_MANUAL (v_ParamListVal[0].Value); // @sic R5-106675 sic@
      }
      case (SELECT_CSG) { // @sic R5-106675 sic@
        v_SystemCmd := cs_MMI_CSG_PLMN (v_ParamListVal);
      }
      case (SELECT_UTRAN_CSG) { //WA#6_3_7
        v_SystemCmd := cs_MMI_CSG_PLMN_UTRAN (v_ParamListVal);
      }
      case (REQUEST_PDPCONTEXT)  {
        v_SystemCmd := cs_AT_REQUEST_PDPCONTEXT(v_ParamListVal);
      }
      case (DEACTIVATE_BEARER)  {
        v_SystemCmd := cs_AT_DEACTIVATEBEARER(v_ParamListVal);
      }

<<SKIPPED SECTION>>


2.9 f_UTRAN_AttachRejectCSG
	Object name 
	f_UTRAN_AttachRejectCSG

	Reason for change
	In the current TTCN implementation of this function, parameter p_CSG_Cell is used to receive the LAU REQ, and also to send the ATTACH REJ and LAU REJ respectively.

This is incorrect and needs to be replaced by utran_CellDedicated instead.

	Summary of change
	Replaced p_CSG_Cell with utran_CellDedicated in LAUREQ, LAU REJ and ATTACH REJ messages respectively.

Please see the screenshot for details.

	Source of change
	UTRAN_CSG_Functions.ttcn

	Label
	WA#


Before change: 

	function f_UTRAN_AttachRejectCSG(UTRAN_CellId_Type p_CSG_Cell) runs on UTRAN_PTC
  {
    var RRC_DATA_IND v_RRC_DataInd;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var START_Value v_START_Value_cs := '00000000000000000000'B;
    alt{
      //Receive Attach Request
      [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeCombined))))) -> value v_RRC_DataInd  //@sic R5s140336 sic@
        {
          // Acc to 24.008 the reject message with the cause 'Not authorized for this CSG' shall be integrity protected.
          // Perform authentication and integrity @sic R5s140366 sic@
          v_START_Value_ps := v_RRC_DataInd.ttcn_start;
          f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
          f_UTRAN_GMM_Authentication();
          f_UTRAN_RRC_Security( p_CSG_Cell, true, ps_domain);
          //Send Attach reject with cause value = 25 Not authorized for this CSG
          U_Dc.send(cas_DataReq_PS(p_CSG_Cell,
                                   tsc_RB3,
                                   cs_U_AttachReject(tsc_NotAuthorizedCSG)));
        }
      //Receive any Location Update request
      [] U_Dc.receive(car_InitDirectTransfer(p_CSG_Cell,
                                             tsc_RB3,
                                             cr_U_LocationUpdatingRequest))-> value v_RRC_DataInd
        {
          // Acc to 24.008 the reject message with the cause 'Not authorized for this CSG' shall be integrity protected.
          // Perform authentication and integrity @sic R5s140366 sic@
          f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value_cs);
          f_UTRAN_MM_Authentication();
          f_UTRAN_RRC_Security(p_CSG_Cell, true, cs_domain);
          //Send Reject with case Not Authorized CSG
          U_Dc.send (cas_DataReq_CS(p_CSG_Cell,
                                    tsc_RB3,
                                    cs_U_LocationUpdatingRej(tsc_NotAuthorizedCSG)));
        }
    }
  }


After change: 

	function f_UTRAN_AttachRejectCSG(UTRAN_CellId_Type p_CSG_Cell) runs on UTRAN_PTC
  {
    var RRC_DATA_IND v_RRC_DataInd;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var START_Value v_START_Value_cs := '00000000000000000000'B;
    alt{
      //Receive Attach Request
      [] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_AttachReq(cr_AttachType(?, (tsc_I_AttachTypeGPRSAtt, tsc_I_AttachTypeCombined))))) -> value v_RRC_DataInd  //@sic R5s140336 sic@
        {
          // Acc to 24.008 the reject message with the cause 'Not authorized for this CSG' shall be integrity protected.
          // Perform authentication and integrity @sic R5s140366 sic@
          v_START_Value_ps := v_RRC_DataInd.ttcn_start;
          f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
          f_UTRAN_GMM_Authentication();
          f_UTRAN_RRC_Security( p_CSG_Cell, true, ps_domain);
          //Send Attach reject with cause value = 25 Not authorized for this CSG
          U_Dc.send(cas_DataReq_PS(utran_CellDedicated /*WA#6_3_7 p_CSG_Cell*/,
                                   tsc_RB3,
                                   cs_U_AttachReject(tsc_NotAuthorizedCSG)));
        }
      //Receive any Location Update request
      [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated /*WA#6_3_7 p_CSG_Cell*/,
                                             tsc_RB3,
                                             cr_U_LocationUpdatingRequest))-> value v_RRC_DataInd
        {
          // Acc to 24.008 the reject message with the cause 'Not authorized for this CSG' shall be integrity protected.
          // Perform authentication and integrity @sic R5s140366 sic@
          f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value_cs);
          f_UTRAN_MM_Authentication();
          f_UTRAN_RRC_Security(p_CSG_Cell, true, cs_domain);
          //Send Reject with case Not Authorized CSG
          U_Dc.send (cas_DataReq_CS(utran_CellDedicated /*WA#6_3_7 p_CSG_Cell*/,
                                    tsc_RB3,
                                    cs_U_LocationUpdatingRej(tsc_NotAuthorizedCSG)));
        }
    }
  }


